

Supplementary Material for Dynamic Structural Equation Modeling with Cycles

Bengt Muthén & Tihomir Asparouhov
Mplus

Loes Keijsers
Erasmus University Rotterdam

April 16, 2024

1 Introduction

Section 2 provides technical background for the definition of amplitude and phase used in the paper. Section 3 provides Mplus inputs for key analyses presented in the paper.

2 Considerations in defining amplitude and phase

Consider the cyclical curve

$$F(t) = A \cos(2\pi w(t - \phi)) = \tag{1}$$

$$A \sin(2\pi w\phi) \sin(2\pi wt) + A \cos(2\pi w\phi) \cos(2\pi wt) = \tag{2}$$

$$\beta_1 x_{1t} + \beta_2 x_{2t} \tag{3}$$

where

$$\beta_1 = A \sin(2\pi w\phi) \tag{4}$$

$$\beta_2 = A \cos(2\pi w\phi) \tag{5}$$

$$x_{1t} = \sin(2\pi wt) \tag{6}$$

$$x_{2t} = \cos(2\pi wt) \tag{7}$$

Generally we are interested in fitting observed data $Y(t)$ to the above curve, meaning that x_{1t} and x_{2t} are the predictors in a linear regression and β_1 and β_2 are the regression coefficients in that linear regression

$$Y(t) = \nu + \beta_1 x_{1t} + \beta_2 x_{2t} + \varepsilon_t. \tag{8}$$

The parameters β_1 and β_2 are uniquely identified from the data, just because the above equation is a standard linear regression. Equations (4-5) can then in principle be used to identify the parameters A and ϕ . These equations however do not identify A and ϕ uniquely. Let's first consider the identification of A . From (4-5) we obtain

$$\beta_1^2 + \beta_2^2 = A^2((\sin(2\pi w\phi))^2 + (\cos(2\pi w\phi))^2) = A^2 \tag{9}$$

$$A = \pm \sqrt{\beta_1^2 + \beta_2^2}. \tag{10}$$

Thus A is uniquely identified by β_1 and β_2 , except that the sign of A is not identified. Furthermore

$$F(t) = A \cos(2\pi w(t - \phi)) = (-A) \cos(\pi + 2\pi w(t - \phi)) = \tag{11}$$

$$(-A) \cos(2\pi w(t - (\phi - 0.5/w))). \tag{12}$$

Thus, the cyclical curve $F(t)$ with parameters A and ϕ is identical to the cyclical curve with parameters $-A$ and $\phi - 0.5/w$. To identify A uniquely it is necessary to impose a constraint on A . We impose the constraint that A is positive. With that constraint, A is uniquely identified and

$$A = \sqrt{\beta_1^2 + \beta_2^2}. \tag{13}$$

Because the range of $F(t)$ is $[-A, A]$, the interpretation of A is that it is half the range of $F(t)$, and is generally referred to as the amplitude of $F(t)$.

Next we focus on the interpretation and the identification of ϕ . Because the cos function has a period of 2π the function $F(t)$ has a period of $1/w$ and

$$F(t) = A \cos(2\pi w(t - \phi)) = A \cos(2\pi w(t - (\phi + n/w))) \quad (14)$$

for any integer n . Therefore the cyclical curve with parameters A and ϕ is identical to the cyclical curve with parameters A and any of the infinitely many number options for the second parameter $\phi \pm 1/w, \phi \pm 2/w, \phi \pm 3/w$, etc. To identify the ϕ parameter we therefore must constrain the parameter to be in the interval of $[0, 1/w)$. Because the cos function has a maximum at 0, the function $F(t)$ has a maximum at ϕ . Therefore with the identification condition that $0 \leq \phi < 1/w$, we can interpret ϕ as the first peak of $F(t)$ after (and including) 0, and we will refer to ϕ as the phase of the curve.

Now we focus on the actual derivation of ϕ . Using (4-5) we see that

$$\frac{\beta_1}{\beta_2} = \tan(2\pi w\phi) \quad (15)$$

The above equation almost precisely identifies the value of ϕ . The tan function has a period of π and therefore the function $\tan(2\pi w\phi)$ has a period of $1/(2w)$. Therefore there are precisely two values of ϕ in the interval $[0, 1/w)$ satisfying equation (15). One, let's call it ϕ_1 , is in the interval $[0, 1/(2w))$ and one, let's call it ϕ_2 , is in the interval $[1/(2w), 1/w)$. Because of the periodicity $\phi_2 = \phi_1 + 1/(2w)$. Note also that

$$\sin(2\pi w\phi_1) = -\sin(2\pi w\phi_2) \quad (16)$$

$$\cos(2\pi w\phi_1) = -\cos(2\pi w\phi_2). \quad (17)$$

Thus for one of the two solutions of (15)

$$A \sin(2\pi w\phi) = \beta_1 \quad (18)$$

$$A \cos(2\pi w\phi) = \beta_2 \quad (19)$$

and for the other solution

$$A \sin(2\pi w\phi) = -\beta_1 \quad (20)$$

$$A \cos(2\pi w\phi) = -\beta_2. \quad (21)$$

Thus equation (15) has two solutions in the interval $[0, 1/w)$, but only one of them satisfies (4-5). Furthermore, because the distance between ϕ_1 and ϕ_2 is precisely half of the period of (1), one of these two values is the highest value of $F(t)$ and the other is the lowest value of $F(t)$. To figure out which one is the correct solution to (4-5) we must take into account the sign of β_1 . Since $0 \leq \phi_1 < 1/(2w)$, $\sin(2\pi w\phi_1) \geq 0$. Therefore if $\beta_1 \geq 0$ the solution to (4-5) is ϕ_1 and if $\beta_1 < 0$ the solution to (4-5) is $\phi_2 = \phi_1 + 1/(2w)$.

Next we connect ϕ_1 and ϕ_2 with the solution ϕ_0 of equation (15) of obtained from the inverse tangent function

$$2\pi w\phi_0 = \arctan\left(\frac{\beta_1}{\beta_2}\right) \quad (22)$$

or equivalently

$$\phi_0 = \frac{1}{2\pi w} \arctan\left(\frac{\beta_1}{\beta_2}\right). \quad (23)$$

The inverse tan function has a range $(-\pi/2, \pi/2)$, and therefore ϕ_0 is in the interval $(-1/(4w), 1/(4w))$. This interval has length of $1/(2w)$. If β_1 and β_2 have the same sign then ϕ_0 is in the interval $[0, 1/(4w))$ and therefore it must be the same as ϕ_1 . If β_1 and β_2 have different signs then ϕ_0 is in the interval $(-1/(4w), 0)$. Since the period of tan in equation (15) is $1/(2w)$, ϕ_0 as well as $\phi_0 + 1/(2w)$ are solutions of equation (15). Since $\phi_0 + 1/(2w)$ is in the interval $(1/(4w), 1/(2w))$, it must be the same as ϕ_1 .

We can now summarize the solution of (4-5) as follows

$$\phi = \begin{cases} \phi_1 = \phi_0 & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 > 0 \\ \phi_1 = \phi_0 + 1/(2w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 < 0 \\ \phi_2 = \phi_1 + 1/(2w) = \phi_0 + 1/(2w) + 1/(2w) & \text{if } \beta_1 < 0 \text{ and } \beta_2 > 0 \\ \phi_2 = \phi_1 + 1/(2w) = \phi_0 + 1/(2w) & \text{if } \beta_1 < 0 \text{ and } \beta_2 < 0 \end{cases} \quad (24)$$

Therefore

$$\phi = \begin{cases} \arctan(\beta_1/\beta_2)/(2\pi w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 > 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/(2w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 < 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/w & \text{if } \beta_1 < 0 \text{ and } \beta_2 > 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/(2w) & \text{if } \beta_1 < 0 \text{ and } \beta_2 < 0 \end{cases} \quad (25)$$

To include the possibility that $\beta_2 = 0$ we provide the complete description of the first peak

$$\phi = \begin{cases} \arctan(\beta_1/\beta_2)/(2\pi w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 > 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/(2w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 < 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/w & \text{if } \beta_1 < 0 \text{ and } \beta_2 > 0 \\ \arctan(\beta_1/\beta_2)/(2\pi w) + 1/(2w) & \text{if } \beta_1 < 0 \text{ and } \beta_2 < 0 \\ 1/(4w) & \text{if } \beta_1 \geq 0 \text{ and } \beta_2 = 0 \\ 3/(4w) & \text{if } \beta_1 < 0 \text{ and } \beta_2 = 0 \end{cases} \quad (26)$$

The definition of phase given above as the first peak after zero is somewhat different than other definitions used in the literature. For example, the cos argument in formula (1) is sometimes specified as $2\pi wt + \phi$ instead of $2\pi w(t - \phi)$. This change amounts to specifying ϕ on a different scale and opposite in sign. While such a change seems trivial and amounts to only a rescaling of the variable, it is quite beneficial for practical purposes. The change in the sign allows us to interpret ϕ as the first peak after 0, which is practical and meaningful and it can be used to model the timing of the daily peaking of the variable. The change in the scale, i.e., the multiplying of ϕ by $2\pi w$, is needed to put ϕ on the interpretable scale of t (which is typically hours in the day or the time scale of the day which can be 3 hour periods). If we do not multiply ϕ by $2\pi w$, the scale of ϕ will be radians, which is impractical. While

other parameterizations of ϕ may yield similar statistical inference, the definition given in formula (1) allows us to connect ϕ directly to the time series plots of the variable.

In addition, it is important to provide a precise computation of ϕ as the first peak after 0, i.e. formula (26), instead of a simple computation of ϕ as

$$\phi = \arctan(\beta_1/\beta_2), \tag{27}$$

which is often recommended as the computation of the phase. This is particularly important for multilevel time-series analysis where we have a population of people and the random β_i coefficients switch signs, not just across individuals but also in the posterior distribution of a single individual. If there is enough variation in the β_1 and β_2 distributions across individuals and within individual, the simple formula (27) will alternate between these 4 values: peak of curve in one period, lowest value of curve in one period, peak of curve in another period, lowest value of curve in another period. Any inference therefore regarding the peaking would be unreliable if we use (27). The same phenomena can be observed also when β_1 and β_2 are non random but the signs of these coefficients vary within their posterior distributions. Note also that arctan formula (27) yields negative and positive values, and therefore it does not refer to the first peak after 0.

Given the complex relationship between the first peak and β_i , as shown in (26), it is clear that a one-stage analysis is not quite reliable for peaking inference. For example, if we specify only that ϕ in (1) is a normally distributed random effect, that would not be enough. The random effect or parameter ϕ must be constrained to the interval $[0, 1/w)$, otherwise the posterior distribution of ϕ may contain the peaks from multiple days which will inevitably be problematic when computing median and mean. This issue might be further complicated when we want to regress the random effect ϕ on other predictors, where the regression parameter posterior distribution may switch signs depending on which period the regression shift points to. Peaks from different periods may be mixed across people and within a single person posterior distribution. This issue is similar to how classes can flip in Mixture models yielding incorrect distributions or how signs of loadings can flip in the factor analysis models. A safer option is to use a two-stage analysis where β_1 and β_2 random effects are estimated and imputed in stage 1, and in stage 2 the random effect computed by (26) is analyzed with multiple imputations.

3 Mplus inputs

Following are Mplus inputs for key runs corresponding to the summary Table 1 of the paper. The DEFINE statements for phase in input 5b are based on the six lines of (26) given above. The inputs for the Monte Carlo simulations are given last. Note that many of the analyses require Mplus Version 8.11 or later.

Table 1: Step 1a: Cross-classified DSEM, Figure 3

	USEVAR = pa; CLUSTER = id time; TINTERVAL = hrs (3 time); LAGGED = pa(1);
ANALYSIS:	TYPE = CROSSCLASSIFIED; ESTIMATOR = BAYES; BITERATIONS = (2000); PROCESSORS = 2;
MODEL:	%WITHIN% pa ON pa&1; %BETWEEN id% pa; [pa] (p0); %BETWEEN time% pa;
OUTPUT:	STANDARDIZED TECH1 TECH4 TECH8;
PLOT:	TYPE = PLOT3; FACTORS = ALL(200);

Table 2: Step 2a: Cross-classified DSEM, Figure 8

	<pre> USEVAR = pa x1 x2; CLUSTER = id time; TINTERVAL = hrs (3 time); BETWEEN = (time) x1 x2; LAGGED= pa(1); </pre>
DEFINE:	<pre> x1 = SIN(6.2831853*(1/8)*time); x2 = COS(6.2831853*(1/8)*time); </pre>
ANALYSIS:	<pre> TYPE = CROSSCLASSIFIED; ESTIMATOR = BAYES; BITERATIONS = (2000); THIN = 10; PROCESSORS = 2; </pre>
MODEL:	<pre> %WITHIN% pa ON pa&1; %BETWEEN id% pa; [pa] (p0); %BETWEEN time% pa ON x1 (p1) x2 (p2); </pre>
MODEL CONSTRAINT:	<pre> LOOP(time,1,56,0.1); PLOT(pacycles fscycles); pacycles = p0 + p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time); fscycles = p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time); </pre>
OUTPUT:	<pre> STANDARDIZED TECH1 TECH4 TECH8; </pre>
PLOT:	<pre> TYPE = PLOT3; FACTORS = ALL(200); </pre>

Table 3: Step 2b: Cross-classified DSEM Factor Analysis, Figure 15

```

USEVAR = PALA1 PALA2 PALA3 PAHA1 PAHA2
PAHA3 x1 x2;
CLUSTER = id time;
TINTERVAL = hrs (3 time);
BETWEEN = (time) x1 x2;

DEFINE:
x1 = SIN(6.2831853*(1/8)*time);
x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:
TYPE = CROSSCLASSIFIED;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10; PROCESSORS = 2;

MODEL:
%WITHIN%
fw1 BY pala1-paha1* (&1 1-4);
fw2 BY paha3* paha2 paha1(&1 11-13);
fw1-fw2@1;
fw1 ON fw1&1; fw2 ON fw2&1;

%BETWEEN id%
fb1 BY pala1-paha1*;
fb2 BY paha3* paha2 paha1;
fb1-fb2@1;

%BETWEEN time%
ft1 BY pala1-paha1* (1-4);
ft2 BY paha3* paha2 paha1 (11-13);
ft1 ON x1 (p11); ft1 ON x2 (p12);
ft2 ON x1 (p21); ft2 ON x2 (p22);

MODEL CONSTRAINT:
LOOP(time,1,56,0.1);
PLOT(ft1cycle ft2cycle);
ft1cycle = p11*SIN(6.2831853*(1/8)*time)+
p12*COS(6.2831853*(1/8)*time);
ft2cycle = p21*SIN(6.2831853*(1/8)*time)+
p22*COS(6.2831853*(1/8)*time);

OUTPUT:
STANDARDIZED TECH1 TECH4 TECH8;

PLOT:
TYPE = PLOT3;
FACTORS = ALL(200);

```

Table 4: Step 2d: Bivariate two-level RDSEM, Figure 7 with a random slope

```
USEVAR = pa tired x1 x2;
CLUSTER = id;
TINTERVAL = hrs (3 time);
WITHIN = x1 x2;
LAGGED = pa(1) tired(1);

DEFINE:      x1 = SIN(6.2831853*(1/8)*time);
             x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:   TYPE = TWOLEVEL RANDOM;
            ESTIMATOR = BAYES;
            BITERATIONS = (2000);
            PROCESSORS = 2;

MODEL:      %WITHIN%
            pa ON x1 x2;
            tired ON x1 x2;
            pa^ ON pa^1;
            tired^ ON tired^1;
            s | pa^ ON tired^;

            %BETWEEN%
            pa tired s WITH pa tired s;

OUTPUT:     STANDARDIZED TECH1 TECH4 TECH8;

PLOT:       TYPE = PLOT3;
            FACTORS = ALL(200);
```

Table 5: Step 3a: Cross-classified DSEM, Figure 8

```

USEVAR = pa x1 x2 sat;
CLUSTER = id time;
TINTERVAL = hrs (3 time);
BETWEEN = (time) x1 x2 sat;
LAGGED = pa(1);

DEFINE:
x1 = SIN(6.2831853*(1/8)*time);
x2 = COS(6.2831853*(1/8)*time);

IF(time>33 .AND. time<40)THEN sat = 1 ELSE
sat=0;

ANALYSIS:
TYPE = CROSSCLASSIFIED;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10; PROCESSORS = 2;

MODEL:
%WITHIN%
pa ON pa&1;

%BETWEEN id%
pa;
[pa] (p0);

%BETWEEN time%
pa ON x1 (p1)
x2 (p2)
sat (p3);

MODEL CONSTRAINT:
LOOP(time,1,56,1);
PLOT(pacycles fscycles saturday patot fstot);
pacycles = p0 + p1*SIN(6.2831853*(1/8)*time)+
p2*COS(6.2831853*(1/8)*time);
fscycles = p1*SIN(6.2831853*(1/8)*time)+
p2*COS(6.2831853*(1/8)*time);
saturday = p3*[34,39];
patot = pacycles + saturday;
fstot = fscycles + saturday;

OUTPUT:
STANDARDIZED TECH1 TECH8;

PLOT:
TYPE = PLOT3; FACTORS = ALL(200);

SAVEDATA:
SAVE = FS(200); FILE = fscyclessat;

```

Table 6: Step 3b: Cross-classified RDSEM, Figure 9

	<pre> USEVAR = pa x1 x2; CLUSTER = id time; TINTERVAL = hrs (3 time); WITHIN = x1 x2; LAGGED = pa(1); </pre>
DEFINE:	<pre> x1 = SIN(6.2831853*(1/8)*time); x2 = COS(6.2831853*(1/8)*time); </pre>
ANALYSIS:	<pre> TYPE = CROSSCLASSIFIED; ESTIMATOR = BAYES; BITERATIONS = (2000); PROCESSORS = 2; </pre>
MODEL:	<pre> %WITHIN% pa ON x1 (p1) x2 (p2); pa^ ON pa^1; %BETWEEN id% pa; [pa] (p0); %BETWEEN time% pa; </pre>
MODEL CONSTRAINT:	<pre> LOOP(time,1,56,0.1); PLOT(pacycles fscycles); pacycles = p0 + p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time); fscycles = p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time); </pre>
OUTPUT:	<pre> STANDARDIZED TECH1 TECH4 TECH8; </pre>
PLOT:	<pre> TYPE = PLOT3; FACTORS = ALL(200); </pre>
SAVEDATA:	<pre> SAVE = FS(200 10); FILE = fscyclesdeviations.dat; </pre>

Table 7: Step 4a: Two-level RDSEM, Figure 6, Cosinor Model

```

USEVAR = pa x1 x2;
CLUSTER= id;
TINTERVAL = hrs (3 time);
WITHIN = x1 x2;
LAGGED = pa(1);

DEFINE:
x1 = SIN(6.2831853*(1/8)*time);
x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:
TYPE = TWOLEVEL RANDOM;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10;
PROCESSORS = 8;

MODEL:
%WITHIN%
sx1 | pa ON x1;
sx2 | pa ON x2;
pa^ ON pa^1;

%BETWEEN%
pa sx1 sx2 WITH pa sx1 sx2;
[pa] (p0);
[sx1] (mx1);
[sx2] (mx2);

MODEL CONSTRAINT:
LOOP(time,1,56,0.1);
PLOT(fscycle pacycle);
fscycle =
mx1*SIN(6.2831853*(1/8)*time)+
mx2*COS(6.2831853*(1/8)*time);
pacycle = p0 +
mx1*SIN(6.2831853*(1/8)*time)+
mx2*COS(6.2831853*(1/8)*time);

OUTPUT:
STANDARDIZED TECH1 TECH4 TECH8;

PLOT:
TYPE = PLOT3;
FACTORS = ALL(200);

```

Table 8: Step 4c: Cross-classified RDSEM, Figure 9 extended to random coefficients

```

USEVAR = pa x1 x2;
CLUSTER = id time;
TINTERVAL = hrs (3 time);
WITHIN = x1 x2;
LAGGED = pa(1);

DEFINE:
x1 = SIN(6.2831853*(1/8)*time);
x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:
TYPE = CROSSCLASSIFIED RANDOM;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10;
PROCESSORS = 2;

MODEL:
%WITHIN%
s1 | pa ON x1;
s2 | pa ON x2;
pa^ ON pa^1;

%BETWEEN id%
pa s1 s2 WITH pa s1 s2;
[pa] (p0);
[s1] (p1);
[s2] (p2);

%BETWEEN time%
pa;

MODEL CONSTRAINT:
LOOP(time,1,56,0.1);
PLOT(pacycles fscycles);
pacycles = p0 +
p1*SIN(6.2831853*(1/8)*time)+
p2*COS(6.2831853*(1/8)*time);
fscycles =
p1*SIN(6.2831853*(1/8)*time)+
p2*COS(6.2831853*(1/8)*time);

OUTPUT:
TECH1 TECH4 TECH8;

PLOT:
TYPE = PLOT3;
FACTORS = ALL(200);

```

Table 9: Step 5a: Two-level RDSEM, Figure 17 factor analysis

```

USEVAR = PALA1 PALA2 PALA3 PAHA1 PAHA2
PAHA3 age SDQ girl tiredavg x1 x2;
CLUSTER = id;
WITHIN = x1 x2;
BETWEEN = girl sdq age tiredavg;
TINTERVAL = hrs (3 time);

DEFINE:      tiredavg = CLUSTER_MEAN(tired);
             girl = sexAA - 1;
             x1 = SIN(6.2831853*(1/8)*time);
             x2 = COS(6.2831853*(1/8)*time);

             CENTER sdq age tiredavg (GRANDMEAN);

ANALYSIS:    TYPE = TWOLEVEL RANDOM;
             ESTIMATOR = BAYES;
             BITERATIONS = (25000);
             PROCESSORS = 2;

MODEL:       %WITHIN%
             fpa1 BY pala1-paha1* (&1 1-4);
             fpa2 BY paha3* paha2 paha1(&1 11-13);
             fpa1-fpa2@1;
             s11 | fpa1 ON x1;
             s12 | fpa1 ON x2;
             s21 | fpa2 ON x1 ;
             s22 | fpa2 ON x2 ;
             fpa1^fpa2 ON fpa1^1 fpa2^1;

             %BETWEEN%
             f1b BY pala1-paha1*;
             f2b BY paha3* paha2 paha1;
             f1b-f2b@1; f1b WITH f2b (c);
             f1b f2b s11-s22 ON girl sdq age tiredavg;

OUTPUT:      STANDARDIZED TECH1 TECH4 TECH8;

MODEL PRIORS: c~ IW(0,3);

PLOT:        TYPE = PLOT3; FACTORS = ALL;

SAVEDATA:    SAVE = FSCORES(200);
             FACTORS = ALL;
             FILE = fscov imp*.dat;

```

Table 10: Step 5b: Analysis of amplitude and phase based on 200 multiple imputations of step 5a

```

DATA:          FILE = fscov implist.dat;
               TYPE = IMPUTATION;

VARIABLE:     NAMES = AGE SDQ GIRL TIREDAVG F1B F2B S11
               S12 S21 S22 B.PALA1 B.PALA2 B.PALA3 B.PAHA1
               B.PAHA2 B.PAHA3 ID;
               USEV = age-tiredavg f1b f2b amp1 amp2 phase1 phase2;
               MISSING = *;

DEFINE:       amp1 = SQRT(s11^2 + s12^2);
               amp2 = SQRT(s21^2 + s22^2);

               IF (s11>=0 .AND. s12>0) THEN phase1 =
               (ATAN(s11/s12))/(6.28*(1/8));
               IF (s11<0 .AND. s12>0) THEN phase1 =
               (6.28+ATAN(s11/s12))/(6.28*(1/8));
               IF (s11>=0 .AND. s12<0) THEN phase1 =
               (3.14+ATAN(s11/s12))/(6.28*(1/8));
               IF (s11<0 .AND. s12<0) THEN phase1 =
               (3.14+ATAN(s11/s12))/(6.28*(1/8));
               IF (s11>=0 .AND. s12==0) THEN phase1 =
               3.14*0.5/(6.28*(1/8));
               IF (s11<0 .AND. s12==0) THEN phase1 =
               3.14*1.5/(6.28*(1/8));

               IF (s21>=0 .AND. s22>0) THEN phase2 =
               (ATAN(s21/s22))/(6.28*(1/8));
               IF (s21<0 .AND. s22>0) THEN phase2 =
               (6.28+ATAN(s21/s22))/(6.28*(1/8));
               IF (s21>=0 .AND. s22<0) THEN phase2 =
               (3.14+ATAN(s21/s22))/(6.28*(1/8));
               IF (s21<0 .AND. s22<0) THEN phase2 =
               (3.14+ATAN(s21/s22))/(6.28*(1/8));
               IF (s21>=0 .AND. s22==0) THEN phase2 =
               3.14*0.5/(6.28*(1/8));
               IF (s21<0 .AND. s22==0) THEN phase2 =
               3.14*1.5/(6.28*(1/8));

               CENTER age sdq tiredavg (GRANDMEAN);

```

Table continues on the next page

Table 11: Step 5b: Analysis of amplitude and phase based on 200 multiple imputations of step 5a, continued

ANALYSIS:	ESTIMATOR = BAYES; FBITERATIONS = 2000;
MODEL:	f1b f2b amp1 amp2 phase1 phase2 on age-tiredavg; [amp1] (a1); [amp2] (a2); [phase1] (p1); [phase2] (p2);
MODEL CONSTRAINT:	NEW(diffAmp diffPh); diffAmp = a1-a2; diffPh = p1-p2;
OUTPUT:	STANDARDIZED TECH4;
PLOT:	TYPE = PLOT3;

Table 12: Monte Carlo simulation using two-level random RDSEM cycles analysis with $N = 200$, $T = 56$. Step 1

MONTECARLO:	NAMES = pa sx1 sx2; NOBSERVATIONS = 11200; NREPS = 500; CSIZES = 200(56); NCSIZE = 1; LAGGED = pa(1); REPSAVE = ALL; SAVE = pa2LRandomstep1T=56Rep=500rep*.dat; BETWEEN = sx1 sx2;
ANALYSIS:	TYPE = TWOLEVEL; ESTIMATOR = BAYES; BITERATIONS = (200); ! complete convergence ! not needed PROCESSORS = 2;
MODEL POPULATION:	%WITHIN% pa^ ON pa^1*0.37243; pa*0.51090; %BETWEEN% pa WITH sx1*-0.00509; pa WITH sx2*-0.01340; sx1 WITH sx2*-0.00127; [pa*5.67306] ; [sx1*-0.08903] ; [sx2*-0.00674] ; pa*0.74775; sx1*0.01524; sx2*0.00752;
MODEL:	Same as MODEL POPULATION

Table 13: Monte Carlo simulation using two-level random RDSEM cycles analysis with $N = 200$, $T = 56$. Step 2

```

DATA:                FILE = pa2LRandomstep1T=56rep=500replist.dat;
                    TYPE = MONTECARLO;

VARIABLE:            NAMES = sx1 sx2 pa id time pa1;
                    USEV = pa x1 x2;
                    CLUSTER = id;
                    LAGGED = pa(1);
                    WITHIN = x1 x2;

DEFINE:              x1 = SIN(6.2831853*(1/8)*time);
                    x2 = COS(6.2831853*(1/8)*time);
                    pa = sx1*x1+ sx2*x2 + pa;

ANALYSIS:            TYPE = TWOLEVEL RANDOM;
                    ESTIMATOR = BAYES;
                    BITERATIONS = (1000);
                    PROCESSORS = 2;

MODEL:               %WITHIN%

                    sx1 | pa ON x1;
                    sx2 | pa ON x2;

                    pa^ ON pa^1*0.37243;

                    pa*0.51090;

                    %BETWEEN%

                    pa WITH sx1*-0.00509;
                    pa WITH sx2*-0.01340;
                    sx1 WITH sx2*-0.00127;

                    [ pa*5.67306 ] ;
                    [ sx1*-0.08903 ] ;
                    [ sx2*-0.00674 ] ;

                    pa*0.74775;
                    sx1*0.01524;
                    sx2*0.00752;

```

Table 14: Monte Carlo simulation using bivariate two-level RDSEM cycles analysis with N = 200, T = 56. Step 1

MONTECARLO:	NAMES = pa tired; NOBSERVATIONS = 11200; NREPS = 500; CSIZES = 200(56); NCSIZE = 1; LAGGED = pa(1) tired1); REPSAVE = ALL; SAVE = pa2LBivstep1T=56Rep=500rep*.dat;
ANALYSIS:	TYPE = TWOLEVEL; ESTIMATOR = BAYES; BITERATIONS = (200); ! complete convergence ! not needed PROCESSORS = 2;
MODEL POPULATION:	%WITHIN% pa^ ON pa^1*0.35281; pa^ ON tired^*-0.12401; tired^ ON tired^1*0.38683; pa*0.49183; tired*1.41098; %BETWEEN% pa WITH tired*-0.53406; [pa*5.66681]; [tired*3.55581]; pa*0.74610; tired*1.43987;
MODEL:	Same as MODEL POPULATION

Table 15: Monte Carlo simulation using bivariate two-level RDSEM cycles analysis with $N = 200$, $T = 56$. Step 2

```

DATA:                                FILE = pa2LBivstep1T=56rep=500replist.dat;
                                     TYPE = MONTECARLO;

VARIABLE:                             NAMES = pa tired id time pa0 pa1 tired0 tired1;
                                     USEV = pa tired x1 x2;
                                     CLUSTER = id;
                                     LAGGED = pa(1) tired(1);
                                     WITHIN = x1 x2;

DEFINE:                                x1 = SIN(6.2831853*(1/8)*time);
                                     x2 = COS(6.2831853*(1/8)*time);
                                     pa = -0.09278*x1-0.01485*x2 + pa;
                                     tired = -0.03656*x1 + 0.52587*x2 + tired;

ANALYSIS:                             TYPE = TWOLEVEL;
                                     ESTIMATOR = BAYES;
                                     BITERATIONS = (1000);
                                     PROCESSORS = 2;

MODEL:                                %WITHIN%

                                     pa ON x1*-0.09278;
                                     pa ON x2*-0.01485;
                                     pa^ ON pa^1*0.35281;
                                     pa^ ON tired^*-0.12401;
                                     tired ON x1*-0.03656;
                                     tired ON x2*0.52587;
                                     tired^ ON tired^1*0.38683;

                                     pa*0.49183;
                                     tired*1.41098;

                                     %BETWEEN%

                                     pa WITH tired*-0.53406;

                                     [ pa*5.66681 ];
                                     [ tired*3.55581 ];

                                     pa*0.74610;
                                     tired*1.43987;

```

Table 16: Monte Carlo simulation using cross-classified DSEM cycles analysis with $N = 200$, $T = 56$. Step 1

```

MONTECARLO:      NAMES = y;
                  NOBSERVATIONS = 11200;
                  NREPS = 500;
                  CSIZES = 200[56(1)]; ! 200 subjects (2b),
                  ! 56 time points (2a)
                  NCSIZE = 1[1];
                  LAGGED = y(1);
                  REPSAVE = ALL;
                  SAVE = pacstep1T=56Rep=500rep*.dat;

ANALYSIS:        TYPE = CROSSCLASSIFIED;
                  ESTIMATOR = BAYES;
                  BITERATIONS = (200); ! complete convergence
                  ! not needed
                  PROCESSORS = 2;

MODEL POPULATION: %WITHIN%
                  y ON y&1*0.371;
                  y*0.513;

                  %BETWEEN LEVEL2A% ! between time
                  y*0.006;

                  %BETWEEN LEVEL2B% ! between individuals
                  y*0.740; [y*5.676];

MODEL:           %WITHIN%
                  y ON y&1*0.371;
                  y*0.513;

                  %BETWEEN LEVEL2A% ! between time
                  y*0.006;

                  %BETWEEN LEVEL2B% ! between individuals
                  y*0.740; [y*5.676];

```

Table 17: Monte Carlo simulation using cross-classified DSEM cycles analysis with $N = 200$, $T = 56$. Step 2

DATA:	FILE = paccstep1T=56Rep=500replist.dat; TYPE = MONTECARLO;
VARIABLE:	NAMES = y time id y1; USEV = y x1 x2; CLUSTER = id time; LAGGED = y(1); BETWEEN = (time) x1 x2;
DEFINE:	x1 = SIN(6.2831853*(1/8)*time); x2 = COS(6.2831853*(1/8)*time); y = -0.088*x1-0.009*x2 + y;
ANALYSIS:	TYPE = CROSSCLASSIFIED; ESTIMATOR = BAYES; BITERATIONS = (1000); PROCESSORS = 2;
MODEL:	%WITHIN% y ON y&1*0.371; y*0.513; %BETWEEN id% y*0.740; [y*5.676]; %BETWEEN time% y*0.006; y ON x1*-0.088 x2*-0.009;
